

## Programmer l'algorithme d'Euclide avec un tableur

Le tableur permet d'afficher clairement, si on le souhaite, les différentes étapes de l'algorithme. Le travail présenté ici se propose de bâtir une feuille de calcul « interactive », c'est-à-dire l'utilisateur et n'affiche que ce qui est utile.

Il est nécessaire de connaître certaines fonctions comme la fonction Si et sa syntaxe :

SI(valeur de test ; action si test positif ; action si test négatif).

Mise en œuvre

1) Commencer par mettre en forme le début de la feuille :

|   | A         | B | C        | D | E        | F  | G     |
|---|-----------|---|----------|---|----------|----|-------|
| 1 | PGCD      | ( |          | ; |          | )= |       |
| 2 | dividende | = | diviseur | × | quotient | +  | reste |

On entre les deux entiers non nuls en C1 et E1.

2) En A3, on reportera la plus grande des deux valeurs absolues des entiers proposés et en C3 la plus petite. On fait en sorte que la feuille n'affiche rien si les deux cases C1 et E1 ne sont pas complètement renseignées.

=SI(ou(estvide(C1);estvide(E1));" ";max(abs(C1);abs(E1)))

Renvoie VRAI si au moins l'une des deux cases C1 ou E1 est vide

Valeur Absolue

- En C3, la même chose en remplaçant max par min. Ainsi l'utilisateur ne se préoccupe ni du signe, ni de l'ordre des valeurs.

- On calcule ensuite le quotient et le reste dans la division euclidienne de A3 par C3. En entre :

=Si(et(estnum(A3);estnum(C3));ent(A3/C3);" ")

- En G3 on entre :

=Si(et(estnum(A3);estnum(C3));mod(A3;C3);" ")

La fonction estnum est une fonction booléenne qui renvoie VRAI si la case testée contient un nombre.

La fonction ent(a/b) renvoie la partie entière de la division a/b.

La fonction mod(a;b), renvoie le reste dans la division euclidienne de a par b.

Les cases B3, D3 et F3 n'affichent rien si la case A3 n'est pas un nombre ; donc, on a :

=Si(estnum(A3);"=";" ") en B3,

et on remplace = par × en D3 et + en F3.

3) Il reste à écrire la ligne 4, car, à ce stade, une recopie des formules vers le bas entraînerait une erreur due à l'utilisation de la ligne 2 qui ne contient pas de nombre.

Pour remplir A4 et C4, on doit d'abord tester si A3 et C3 sont bien numériques, puis si G3 (le reste précédent) est non nul.

Si ces trois conditions sont remplies, on place C3 en A4 et G3 en C4.

- En A4, on a donc :

=SI(et(estnum(A3);estnum(C3);G3<>0);C3;" ")

Adapter la formule pour C4.

Les calculs du quotient et du reste sont les mêmes ; donc on peut, pour les colonnes B, D, E, F et G recopier les formules vers le bas jusqu'à la ligne 3.

4) On sélectionne la plage A4 : G4 et on copie vers le bas jusqu'à un nombre suffisant de lignes, 80 par exemple (nettement suffisant, il est rare d'avoir autant de lignes pour l'algorithme d'EUCLIDE !).

5) Il reste maintenant à calculer le PGCD. On va utiliser la colonne H.

Le PGCD est le dernier reste non nul, donc, c'est le diviseur dans la ligne où le reste est nul.

• En H3, on entre :

=SI(OU(ESTVIDE(A3);ESTVIDE(C3));" ";SI(G3=0;C3;" "))

puis on tire vers le bas jusqu'à la ligne 80.

• On masque ensuite cette colonne H en la sélectionnant, puis en utilisant le menu correspondant, par le clic droit de la souris.

6) Enfin en G1, on entre :

=SI(OU(ESTVIDE(C1);ESTVIDE(E1));" ";MAX(H3:H80))

On va ainsi rechercher la seule valeur non nulle de la colonne, c'est-à-dire le PGCD. On peut mettre cette case en gras et dans une taille supérieure au reste pour que le résultat saute aux yeux.

Applications

1) Donner à a et b les valeurs 45 212 et 30 148.

| G1 |           | f_x |          | =SI(OU(ESTVIDE(C1);ESTVIDE(E1));" ";MAX(H3:H80)) |          |    |          |   |   |  |
|----|-----------|-----|----------|--|----------|----|----------|---|---|--|
|    | A         | B   | C        | D  | E        | F  | G        | H | I |  |
| 1  | PGCD      | (   | 45212    | ;  | 20148    | )= | <b>4</b> |   |   |  |
| 2  | dividende | =   | diviseur | ×  | quotient | +  | reste    |   |   |  |
| 3  | 45212     | =   | 20148    | ×  | 2        | +  | 4916     |   |   |  |
| 4  | 20148     | =   | 4916     | ×  | 4        | +  | 484      |   |   |  |
| 5  | 4916      | =   | 484      | ×  | 10       | +  | 76       |   |   |  |
| 6  | 484       | =   | 76       | ×  | 6        | +  | 28       |   |   |  |
| 7  | 76        | =   | 28       | ×  | 2        | +  | 20       |   |   |  |
| 8  | 28        | =   | 20       | ×  | 1        | +  | 8        |   |   |  |
| 9  | 20        | =   | 8        | ×  | 2        | +  | 4        |   |   |  |
| 10 | 8         | =   | 4        | ×  | 2        | +  | 0        | 4 |   |  |
| 11 |           |     |          |  |          |    |          |   |   |  |
| 12 |           |     |          |  |          |    |          |   |   |  |

Pour plus de sécurité, on peut ensuite protéger la feuille contre toute mauvaise manipulation. Sélectionner la feuille, puis dans le menu format-cellule, verrouiller la protection.

Ensuite, sélectionner C1 et E1 et ôter leur protection.

Enfin, dans le menu outils-protection, sélectionner protéger la feuille (le mot de passe n'est pas utile).

2) On peut de même bâtir une feuille permettant, à l'aide de cet algorithme, de trouver un couple solution dans  $\mathbb{Z}^2$  pour l'équation :

$$au + bv = \text{PGCD}(a;b).$$